
ABSTRACT 
When you develop a SAS/AF® application, you have to be ready to handle the worst. Files may be unavailable, users 
may provide unexpected input, or a change in configuration may catch you unaware—your program needs to behave 
properly under any circumstances. 

SAS Component Language now provides facilities for handling exceptions and for trapping run-time errors 
which allow you to make your applications much more robust. This paper discusses how to use these tools to make 
your application impervious to the slings and arrows of fate. 

INTRODUCTION 
Making full-fledged interactive SAS application behave well can be a tricky business. You can never be sure under 
exactly what circumstances your code will run, and a violation of your assumptions can cause some part of your code 
to crash.  A necessary file may be missing or locked; a user might accidentally try to perform operations in the wrong 
sequence or provide an inconsistent specification; or you as the developer may have incorrectly assumed that a 
variable name or type would not change. In any of these cases, you need some way to detect and handle these 
conditions while your application is running. 

Fortunately, SAS/AF provides tools which can help us make an application ready for the unexpected. 
Exception handling and program halt components provide a way to take greater control of an application. 

EXCEPTIONS 
Exception handling is found in Java and many other programming languages, and was added to SAS Component 
Language with SAS version 8. Exceptions provide a way for a developer to control the flow of an application when an 
unexpected situation or an error is encountered.  

WHAT CAN GO WRONG? 
Even the simplest operation provides many opportunities for something to go wrong when the program is run. For 
example, consider a program that opens a SAS table and recovers a value stored in the first row: 

do ; 
   declare num myDSID fetchRC ; 
   myDSID=open('SASUser.myDataSet', 'i') ; 
   fetchRC=fetch(myDSID) ; 
   theVal=getVarN(myDSID, varnum(myDSID, 'aNumVar')) ; 
   myDSID=close(myDSID) ; 
  end ; 

This code makes several assumptions: 
• The specified data set SASUser.myDataSet exists, and can be opened by the user. 
• The data set has at least one observation to be returned by the FETCH() function. 
• The data set contains a numeric variable named aNumVar. 

Some of these assumptions make sense in most cases—if the application has just created the data set, for example, 
you know the structure—but even in those cases problems can arise. If the SASUser library points to a network drive, 
or if you are using a library which points to a SAS/SHARE® server, then a server or network error can prevent the 
application from opening the data set even if the file was properly created. 
When you add user input and configuration, assumptions are even more dangerous. Even validated user input could 
introduce problems not accounted for by your validation routine. 

CHECKING RETURN CODES 
An obvious first step is to use the return codes from functions like OPEN() and FETCH() to check some of these 
assumptions. However, checking the return codes at each phase and proceeding conditionally produces a deeply 
nested flow of control: 

do ; 
   myDSID=open('SASUser.myDataSet', 'i') ; 
   if myDSID = 0 then do ; 
      declare char theMsg=SysMsg(); 
      put 'Error opening SASUser.MyDataSet' ; 
      put theMsg ; 
     end ; 
   else do ; 
      if fetch(myDSID) ne 0 then  

         put 'No observation retrieved from SASUser.MyDataSet' ; 
      else theVal=getVarN(myDSID, variablePosition) ; 
      myDSID=close(myDSID); 
     end ; 
  end ; 

Note that this structure adds an additional level of nesting for every return code that we check—and in this case, we 
haven’t bothered to check whether the data set actually contains a numeric variable named ANumVar.  
FLOW CONTROL WITH EXCEPTIONS 
By using exceptions, we can control the program flow and avoid all conditional execution. Exceptions are “thrown” and 
“caught.” When you THROW an exception, the program stops executing, looks for a CATCH statement, and executes 
the code that follows the CATCH statement. The code in the CATCH block is not executed unless an exception is 
thrown. 

Using exceptions makes it easier to control the flow of code execution: 
import SASHelp.Classes.SCLException ; 
do ; 
   declare SCLException scle ; 
   declare num myDSID ; 
   myDSID=open('SASUser.MyDataSet', 'i') ; 
   if myDSID=0 then do ; 
      declare SCLException  
         scle=_new_ SCLException( 
            'Error opening SASUser.MyDataSet.  ' || 
            SysMsg()) ; 
      throw scle ; 
     end ; 
   if fetch(myDSID) ne 0 then do ; 
      declare SCLException  
         scle=_new_ SCLException( 
            'No observation retrieved from SASUser.MyDataSet') ; 
      throw scle ; 
     end ; 
   theVal=getVarN(myDSID, varnum(myDSID, 'aNumVar')) ; 
   myDSID=close(myDSID) ; 
   catch scle ; 
      declare char theMessage=scle.getMessage(); 
      put theMessage ; 
   endCatch ; 
  end ; 

This snippet of code is longer than the first example, but has two advantages. First, the conditional structure is more 
straightforward. There are no more ELSE blocks, and the GETVARN() function is no longer nested within two 
conditional loops.  

Second, we now have a single PUT statement that runs for all error conditions. Thus, if we later decide that 
we should notify the user with a MESSAGEBOX() function rather than just logging the error, we only need to make a 
change in one location. 

The above example uses the most basic exception class, SASHelp.Classes.SCLException.class (which is 
a subclass of the abstract class SASHelp.Classes.SCLThrowable.class). The constructor for this method takes a 
single character parameter, which populates the exception’s message attribute. When catching an SCLException, the 
getMessage()  method  returns this message string. The exception itself provides communication between the 
throwing code and the catching code. 

Note the general structure shown here for using exceptions: 
• Enclose the code in a DO… END block. 
• Begin the DO block by declaring the exception so the scope is local to the DO block. 
• End the DO block with the CATCH… ENDCATCH block. 

THE CALL STACK 
Within a single block of SCL code, the flow control provided by exception handling seems unremarkable. However, 
once the SCL code starts to call other class methods, FRAME entries, or SCL programs, you can take advantage of 
the fact that the calling program can catch and handle exceptions thrown by the called program or object. 

For example, say we have a DataReader class whose methods open a SAS table and retrieve the contents. 
If our program calls a method on the DataReader, and the DataReader throws an exception, then the calling program 
can catch and handle the exception. For example, the constructor of a DataReader might open the data set and throw 
an exception if this fails. 

DataReader: public method 

   theDataSet:input:char 
   ; 
   theDSID=open(theDataSet, 'i') ; 
   if theDSID=0 then do ; 
      declare SCLException 
         scle=_new_ SCLException( 
            'There was an error opening the data set ' || 
            theDataSet || '.  SysMsg = ' || SysMsg()); 
      throw scle ; 
     end ; 
 endMethod ; 

The calling code can then depend on the DataReader object to throw the exception: 
do ; 
   declare SCLException scle ; 
   declare DataReader 
      dr=_new_ DataReader('SASUser.MyDataSet', 'i') ; 
 
   [...code to actually read from the data set...] 
 
   dr._term() ; 
    
   catch scle ; 
      declare char theMessage=scle.getMessage() ; 
      put theMessage ; 
   endCatch ; 
  end ; 

In this case, if the data set is unavailable, the DataReader will throw an exception that will be caught by the calling 
code. 

This example suggests another advantage to using exceptions. Without exceptions, an object (like the 
DataReader described) would have to be designed with a set of return codes so the calling program could check for 
success before continuing. By taking advantage of the exception handling structure, the calling program does not 
need to continually verify that the DataReader object has succeeded, and the DataReader object does not have to 
worry about communicating with the calling program when there is a problem.  

CREATING AN EXCEPTION CLASS 
The SCLException class is convenient, but you can gain a great deal more power by creating your own subclass of 
SCLException: 

class SASUser.MyApp.MyAppException.class extends 
   SASHelp.Classes.SCLException.class ; 

One obvious feature to add is some sort of user notification method; this way you can have some standardized 
code—using the SAS LOG, the MESSAGEBOX() function, or even a FRAME entry—to handle user notification every 
time you use the exception. For example, if you define the method NotifyUser() as follows: 

/* notifyUser */ 
notifyUser: public method; 
   * Use hasBeenDisplayed so the user is never warned ; 
   * twice about the same exception ; 
   if hasBeenDisplayed then return ; 
   do ; 
      declare char theButton ; 
      declare list  
         messageList=makeList(); 
      messageList=insertc(messageList, message, -1) ; 
      theButton=messageBox(messageList, 'S', 'O', 'Error', 'O') ; 
      if delList(messageList) then  
         put 'Error deleting message list.' ; 
     end ; 
   hasBeenDisplayed=1 ; 
  endmethod; 

you can call it from a CATCH block when handling the error: 
do ; 
   declare MyAppException mae ; 
   [...code which may throw a MyAppException...] 
   catch mae ; 
      mae.notifyUser(); 
   endCatch ; 

  end ; 
You might also want to create more than one kind of customized exception. For example, you might want a further 
subclass for errors encountered when opening SAS tables that is initialized not only with a message but with the error 
code and error message generated by SAS. 

RULES FOR EXCEPTIONS 
Here are some additional rules and guidelines for using exception handling in your application: 

• If you have multiple classes that are descendents of SCLException, note that a CATCH statement for a 
parent class will catch an instance of the child class. If you want to handle them differently, the CATCH block 
for the child class should come before that for the parent class. 

• If you have an object whose methods might throw exceptions, you should always have a CATCH block 
available when you are calling those methods. Otherwise, an uncaught exception will cause a program halt. 
Don’t throw what you can’t catch. 

• Within a CATCH block, you can THROW the exception again if you want to pass control farther up the 
calling stack. (If your exception has some sort of user notification method, you might want to remember when 
it is called in order to avoid multiple identical warnings resulting from a single error.) 

• If you run a FRAME entry, and the SCL code for that FRAME throws an exception to be caught by the calling 
program, the CATCH block will not execute until control returns to the calling program. It is usually a good 
idea for the frame to have its own exception-handling code and to only throw exceptions for the calling 
program within its TERM block. 

• Because of the way the LINK statement deals with variable scope (in short, not well), LINKing is not 
compatible with exception handling in SCL.  If you have a LINK statement within a DO loop, and the LINKed 
code throws an exception, control will not properly be transferred to the CATCH block within that DO loop.  

HALT HANDLING 
So far we have discussed throwing exceptions when your program knows something is wrong. But what about those 
cases when the program actually crashes? Both the dread message “WARNING: Program halted. See LOG for 
messages” in the status line and the little line in the log that says “Program returning prematurely at line N” are user-
hostile in a production environment and can unnerve even an experienced user. 
This is the default behavior for a run-time error, but there is a better way to handle this situation. SAS provides the 
ProgramHalt class, SASHelp.Classes.ProgramHalt.class, which uses event handlers for run-time errors to allow 
control of the program even after an unexpected error shows up. By creating a subclass of ProgramHalt and 
overriding these methods, you can specify code to run when such an error occurs. 

 

TYPES OF PROGRAM HALT 
There are a number of types of program halt that we can handle by overriding the associated methods in a 
descendant of the ProgramHalt class: 

• Generic error (override the _onGeneric() method): This is probably the most frequent kind of halt. This 
shows up most often when you pass inappropriate parameters to a function. For example, if you try to 
GETVARN() on a position containing a character variable, or GETITEMC() with an index that is greater than 
the length of the list. 

• Attribute error (override the _onAttributeError() method): This happens when you assign an invalid value to 
an object’s attribute. For example, if you assign _frame_.pMenuEntry to a character value that is not the 
name of an existing PMENU entry, you will get an attribute error. 

• Divide by zero error (override the _onZeroDivide() method): This happens when you ask SAS to divide a 
value by zero. 

• Overflow error, underflow error, and generic math error (override the _onOverflow(), _onUnderflow(), and 
_onGenericMath() methods): These errors happen when a SAS calculation fails. 

HALT ATTRIBUTES 
The most important thing we can do in a halt handler method is to decide whether or not to continue. If we set the 
attribute StopExecution to ‘NO’, then SAS continues running and allows us to throw an exception.  
Other attributes allow us to get information on what went wrong: 

• dump contains a list of the error messages generated 
• entry is the four-level name of the entry that was running when the halt occurred 
• lineNumber identifies the line of code running when the halt occurred 
• traceBack is a list containing the calling stack at the time of the halt 

Generic errors have two additional attributes: 
• keywordType indicates whether the error was in a function or a method 
• keyword contains the specific method or function that caused the error 

For example, if you call GETITEMN() on a list position that contains a character value, keywordType is “function” 
and keyword is “GETITEMN”. 

HALTS AND EXCEPTIONS 
What should our application actually do when we encounter a run-time problem? Neither halting nor continuing 
blithely upon our way is ideal--the former gives us no way to recover, and the latter might fail since the program will 
assume that all has gone normally. 

For example, the following implementation of an _onGeneric() method logs all the available information and 
throws an exception: 

/* _onGeneric: Invoked when the generic halt event occurs */ 
onGeneric: public method; 
   declare SCLException scle ; 
    
   * Keep the program running ; 
   stopExecution='NO' ; 
   
   * Put a message explaining how the error occurred ; 
   do ; 
      declare char whatWentWrong ; 
      if keyword ne '' then  
         whatWentWrong='Error in ' || keywordType || ' ' || keyword ; 
      else whatWentWrong='Generic exception' ; 
      put whatWentWrong 'encountered in' entry 'at line' lineNumber; 
     end ; 
      
   * Put the halt message and the stack trace into the log ; 
   call putlist(dump, 'Listing of SCL halt', 2) ; 
   call putlist(traceback, 'Call stack of SCL halt', 2) ; 
    
   scle = _new_ SCLException( 
      'A program halt occurred.  See the SAS log for details.'); 
   throw scle ; 
  endmethod; 

Having the application throw an exception is the ideal solution. This provides a way for the program to keep going, but 
at the same time transfers control to error-handling code that you put in the CATCH block. 
To handle run-time errors, simply instantiate your halt handler class at the start of your application. This creates the 
object, which then sits quietly in the background and assumes control when and if a program halt is encountered. 

A word of warning: if you create a class that handles halts and throws exceptions, you must always have a 
CATCH block to fall back to. An uncaught exception generates a system halt. If this system halt in turn generates an 
exception (which is still not caught), then SAS becomes stuck in a loop. The only solution in this case is to forcibly quit 
SAS. 

CONCLUSION 
Exceptions and the program halt component allow you to make your SAS applications more robust and to prevent 
strange behavior at run time. 

CONTACT INFORMATION 

